# Chapter 5

# Monopoly, Oligopoly and Increasing Returns

Now we turn to models involving imperfect competition and increasing returns to scale. These topics have attracted considerable attention in both theoretical and empirical analyses of international trade and public economics over the last two decades. Imperfect competition models, at least general-equilibrium models used in international trade and in public economics often fall into one of two cases. First, there are oligopoly models with small numbers of firms strategically interacting with one another. These models often assume that an industry produces a homogeneous good. Second, there are monopolistic-competition models with large numbers of firms producing differentiated goods. Individual firms are small relative to the market, which results in greatly simplified marginal revenue functions or market rules.

Similarly, there are several classes of scale economies. An older literature in international trade assumes that scale economies are external to individual firms but internal to industries. This has some appeal empirically, but has some significant analytical advantages in general-equilibrium models in that competitive pricing rules can be used. A newer literature assumes that scale economies are internal to individual firms

so that imperfect competition is an inevitable part of the equilibrium analysis.

Because these models have features rather different from the more standard competitive general-equilibrium models, I will present both MCP and MPS/GE versions of each model so that the reader can see exactly what is solved for in the MPS/GE versions. The models in this chapter are as follows:

M51-MCP.GMS Closed economy model with monopoly in the X sector, MCP version

M51-MPS.GMS Closed economy model with monopoly in the X sector, MPS/GE version

M52-MCP.GMS Closed economy model with monopoly and increasing returns in the X sector, MCP version.

M52-MPS.GMS Closed economy model with monopoly and increasing returns in the X sector, MPS/GE version.

M53-MCP.GMS Closed economy model with monopoly, increasing returns and free entry/exit in the X sector, MCP version

- M53-MPS.GMS Closed economy model with monopoly, increasing returns and free entry/exit in the X sector, MPS/GE version
- M54-MCP.GMS Two country trade model with monopoly, increasing returns and free entry/exit in the X sector, MCP version
- M54-MPS.GMS Two country trade model with monopoly, increasing returns and free entry/exit in the X sector, MPS/GE version

# Model M51-MCP Closed economy, monopoly in the X sector, MCP version

This is a standard two-good, two-factor, closed-economy general-equilibrium model that is very similar to those used in earlier chapters. Indeed, we start with a data matrix that is very similar to those used in earlier chapters. Activities are X, Y and W (welfare or utility). Factors of production in this and in the next chapter are called unskilled and skilled labor. Unskilled labor is typically called L with a price of PW or just W, and skilled labor is called S with a price of PZ or just Z. PU is the price of a unit of utility (the value of the unit expenditure function).

The monopoly or oligopoly markup will be denoted MK or MARKUP. There are two agents, the representative consumer who receives all the factor incomes and tax revenue (if any) and pays subsidies (if any). Then there may be an agent called ENTRE who receives markup revenue and pays fixed costs (if any).

Here is the data matrix for our first monopoly model.

		Pro	Production Sectors			Consumers		
Markets		X	Y	W		CONS	ENTRE	
PX	!	100		-100				
PY			100	-100				
PU	İ			200	j	-180	-20	
PW	İ	-32	-60		j	92		
PZ	İ	-48	-40		İ	88		
MK	j	-20			İ		20	

We assume that 20% of the value of the X output accrues to the agent ENTRE as a monopoly profit. For reasons connected with theory, we specify the markup as a deduction from the consumer price, not as an addition to marginal cost. Denoting marginal cost by mc, the markup is given by the formula

$$p_r(1-mk) = mc$$
  $100(1-mk) = 80 \Rightarrow mk = 0.20$ 

Written in this form, the markup implied by the data matrix is 20%.

It is desirable that the markup have some basis in theory. Therefore, I will take a short digression into economic theory to derive a marginal revenue function and see how

a monopoly markup relates to the underlying elasticity of substitution in preferences and demand.

Suppose demand for good X is just written in inverse form p(X) so the monopolist's revenue is R = p(X)X. Marginal revenue is then given by:

(1) 
$$\frac{\partial R}{\partial X} = p + X \frac{\partial p}{\partial X} = p + p \left[ \frac{X}{p} \frac{\partial p}{\partial X} \right] = p \left[ 1 - \frac{1}{\eta} \right] \equiv MR$$

$$MR = p(1 - mk)$$
  $mk = \frac{1}{\eta}$   $\eta = -\left\lfloor \frac{p}{X} \frac{\partial X}{\partial p} \right\rfloor$ 

where  $\eta$  is the Marshallian elasticity of demand, defined as positive. The monopoly markup is just the inverse of this elasticity. This looks good, but what is  $\eta$  in general equilibrium?

Suppose now that there are two goods,  $X_1$  and  $X_2$ , and consumer income is given by M. I will assume a symmetric CES utility function and assure the reader that the formula I derive is applicable to a more general case with different weights on the goods. The utility function is given as follows, where  $\sigma$  is the elasticity of substitution between goods. For a monopoly equilibrium to exist, we must have  $\sigma > 1$ , which in turn imposes the restriction that  $1 > \alpha > 0$ .

$$(2) \quad U = \left[X_1^{\alpha} + X_2^{\alpha}\right]^{\frac{1}{\alpha}}$$

Maximizing utility subject to the usual linear budget constraint, yields demand functions:

(3) 
$$X_i = p_i^{-\sigma} \left[ \sum p_j^{1-\sigma} \right]^{-1} M$$
  $\sigma = \frac{1}{1-\alpha}, \quad \alpha = \frac{\sigma-1}{\sigma}$ 

Taking the own-derivative, we have:

$$(4) \frac{\partial X_{i}}{\partial p_{i}} = -\sigma p_{i}^{-\sigma-1} \left[ \sum_{j=0}^{\sigma-1} p_{j}^{1-\sigma} \right]^{-1} M + p_{i}^{-\sigma} \left[ \sum_{j=0}^{\sigma-1} p_{j}^{1-\sigma} \right]^{-2} (\sigma-1) p_{i}^{-\sigma} M$$

Forming the elasticity, we get:

(5)

$$\frac{p_{i}}{X_{i}} \frac{\partial X_{i}}{\partial p_{i}} = -\sigma + (\sigma - 1)p_{i}^{1 - \sigma} \left[ \sum p_{j}^{1 - \sigma} \right]^{-1} \qquad s_{i} \equiv \frac{p_{i}X_{i}}{M} = p_{i}^{1 - \sigma} \left[ \sum p_{j}^{1 - \sigma} \right]^{-1}$$

The second equation gives the share of income spent on good i,  $s_i$ , by multiplying through (3) by  $p_i$  and dividing both sides by M. We see that this share appears in the second term of the first equation. Thus the elasticity of demand for  $X_i$  can be written as:

(6) 
$$\eta_i = -\frac{p_i}{X_i} \frac{\partial X_i}{\partial p_i} = \sigma - (\sigma - 1)s_i \qquad mk = [\sigma - (\sigma - 1)s_i]^{-1}$$

Note that Cobb-Douglas,  $\sigma = 1$ , is a special case in which the Marshallian elasticity is also equal to 1.

Now we are able to calibrate our data to an underlying theory. That is, we assume that this data is generated by a standard monopoly pricing model and that preferences are CES. The data show that the share of expenditure on X is 0.50, and the markup is 0.20, so  $\eta = 5$ . Solving for the implied elasticity of substitution in preferences, we get  $\sigma = 9$ . We will use this value in the data to follow.

The final issue is also one of calibration. If we let the marginal cost (or producer price) of X = 1, then the markup implies that the consumer price is  $p_x = 1.25$ . Suppose that we wish to chose units so that all other prices are equal to one, a convention that we like to adopt. The cost function for utility (utility price = pu) which is dual to the utility function given above is given by:

$$(7) \quad pu = \left[p_x^{1-\sigma} + p_y^{1-\sigma}\right]^{\frac{1}{1-\sigma}}$$

In our case, the price of good X is 1.25, so we can divide  $p_x$  in this cost function to

calibrated initially so that both additive terms in the function equal 1. But then the term in square brackets equals 2, so if we want  $p_u = 1$  initially we have to compensate with a multiplicative constant on the front of the function. Denote this term as A. The calibrated cost function for utility is then:

$$A*((PX/1.25)**(1-SIGMA) + PY**(1-SIGMA))**(1/(1-SIGMA))$$

$$=G= PU;$$

$$A = 0.5**(1/(1-SIGMA))$$

This is satisfied with equality at  $p_x = 1.25$ ,  $p_y = 1$ , and  $p_u = 1$ . The parameter A will then also appear in the demand functions for X and Y, which are derived from Shepard's lemma. The demand for X, for example, is the derivative of (7) with respect to px, times the level of welfare. Initial X output is X\*80 (i.e., the activity level X is set at one, where X = 1 generates 80 units of output) and initial welfare is similarly W\*200. Thus the demand for X is given by;

Now you might see why I like MPS/GE (and this is a pretty simple case). The latter will automatically calculate scaling coefficients like A from the data provided in a production block, and then automatically apply Shepard's lemma to get commodity and factor demands.

The rest of the program should be fairly transparent at this point. Two equations are added to calculate the share of expenditure on X (SHX) and the markup (MK). The agent ENTRE receives income from markups and demands utility just like the representative agent. After the model solves, we calculate how total income is divided between the representative factor owner (INCOMEC) and the monopolist (INCOMEM).

The counter-factual experiment is to impose marginal-cost pricing, which we can do by simply fixing the markup at zero with the statement MARKUP.FX = 0. Not only does welfare increase, but factor owners have a much bigger proportional increase since there is also a redistribution from monopoly profits to the factor owners. Overall welfare rises by 4% following the imposition of marginal-cost pricing, so this is a measure of deadweight loss from monopoly. But consumer (factor owner) welfare rises by 15.6% (0.90 to 1.04) which is much more significant.

\$TITLE Model M51-MCP.GMS: Closed 2x2 Economy, monopoly X \* producer MCP version

### \$ONTEXT

		Pro	oduction	n Sectors		Consur	mers
Markets		X	Y	W		CONS	ENTRE
PX	!	100		-100			
PY			100	-100			
PU	ĺ			200	j	-180	-20
PW		-32	-60			92	
PZ		-48	-40		j	88	
MK		-20			ĺ		20

### \$Offtext

#### **PARAMETERS**

A Scale parameter for utility function, SIGMA Elasticity of substitution, INCOMEM Monopoly profit (in welfare units),

```
INCOMEC factor owners' income,
ENDOWS
ENDOWL
MODELSTAT;
SIGMA = 9;
ENDOWS = 88;
ENDOWL = 92;
A = 0.5**(1/(1-SIGMA));
POSITIVE VARIABLES
X
Y
W
PX
PΥ
PU
PZ
PW
CONS
ENTRE
SHAREX
```

#### MARKUP;

```
EQUATIONS
DX
         Demand for X
         Demand for Y
DY
DW Demand for W
PRICEX MR = MC in X
PRICEY
         Zero profit condition for Y (PY = MC)
         Zero profit condition for W
PRICEW
SKLAB
         Supply-demand balance for skilled labor
         Supply-demand balance for unskilled labor
UNLAB
         Consumer (factor owners') income
ICONS
         Entrepreneur's profits
IENTRE
SHX
         Share of X in expenditure
         Markup equation;
MK
PRICEX.. (PW**0.40)*(PZ**0.60) = G = PX*(1 - MARKUP);
PRICEY.. (PW**0.60)*(PZ**0.40) = G = PY;
PRICEW.. A*((PX/1.25)**(1-SIGMA) +
```

```
PY**(1-SIGMA))**(1/(1-SIGMA)) = G = PU;
            X*80 = G = A*(PX/1.25)**(-SIGMA)*
DX..
                   ((PX/1.25)**(1-SIGMA) +
                  PY**(1-SIGMA)) **(SIGMA/(1-SIGMA))
                   *W*200/1.25;
DY..
            Y*100 = E = A*PY**(-SIGMA)*
                   ((PX/1.25)**(1-SIGMA) +
                  PY**(1-SIGMA))**(SIGMA/(1-SIGMA))
                    *W*200:
            W*200 = E = (CONS + ENTRE)/PU;
DW..
SKLAB.. ENDOWS = E = 0.40*(PW**0.60)*(PZ**(0.40-1))*Y*100
                 + 0.60*(PW**0.40)*(PZ**(0.60-1))*X*80;
UNLAB.. ENDOWL = E = 0.60 \times (PW \times (0.60 - 1)) \times (PZ \times (0.40) \times Y \times 100)
                 + 0.40*(PW**(0.40-1))*(PZ**0.60)*X*80;
```

ICONS.. CONS =E = PZ\*ENDOWS + PW\*ENDOWL;

```
IENTRE.. ENTRE =E = MARKUP*PX*X*80;
SHX.. SHAREX =E= 80*PX*X / (80*PX*X + 100*PY*Y);
MK.. MARKUP = E = 1/(SIGMA - (SIGMA-1)*SHAREX);
MODEL M51 /DX.PX, DY.PY, DW.PU, PRICEX.X, PRICEY.Y,
PRICEW.W,
           SKLAB.PZ, UNLAB.PW, ICONS.CONS, IENTRE.ENTRE
           SHX.SHAREX, MK.MARKUP/;
OPTION MCP=MILES;
OPTION LIMROW=0;
OPTION LIMCOL=0;
SOFFSYMLIST OFFSYMXREF OFFUELLIST OFFUELXREF
CONS.L = 200;
X.L = 1;
Y.L = 1;
W.L = 1;
PX.L = 1.25;
```

```
PY.L = 1;
PZ.L = 1;
PW.L = 1;
PU.L = 1;
CONS.L = 180;
ENTRE.L = 20;
SHAREX.L = 0.5;
MARKUP.L = 0.20;
PY.FX = 1;
*M51.ITERLIM = 0;
SOLVE M51 USING MCP;
MODELSTAT = M51.MODELSTAT - 1.;
INCOMEM = W.L*(ENTRE.L/(ENTRE.L + CONS.L));
INCOMEC = W.L*(CONS.L/(ENTRE.L + CONS.L));
DISPLAY INCOMEM, INCOMEC;
```

```
* Evaluate the potential gains from first-best (marginal
* cost) pricing:

MARKUP.FX = 0;

SOLVE M51 USING MCP;

INCOMEM = W.L*(ENTRE.L/(ENTRE.L + CONS.L));
INCOMEC = W.L*(CONS.L/(ENTRE.L + CONS.L));

DISPLAY INCOMEM, INCOMEC;
```

# Model M51-MPS Closed economy, monopoly in X, MPSGE version

This model is identical to the previous model, except that this one is coded in MPS/GE. There is not a lot that needs to be said here. The basic simplifying trick is to use the "N" field in a production block to denote an endogenous "tax rate", which is in this case the markup. The revenue from this tax is assigned to the agent ENTRE as monopoly profits. The endogenous tax rate is an auxiliary variable which is then set in a constraint equation. The production block is given by:

```
$PROD:X s:1
O:PX Q: 80 A:ENTRE N:MARKUP
I:PW Q: 32
I:PZ Q: 48
```

Note that if we use the formula derived above, we want the tax set on the output so that we indeed have PX(1 - MARKUP) = MC. If the tax was set on the two input fields, we would have PX = (1+MARKUP)MC. The same markup does not yield the same relationship between price and marginal cost in the two cases since the tax base is different in the two situations. So be careful to specify the production block carefully.

Second, remember to include the price field in the production block for welfare in

order for MPS/GE to correctly calibrate the function. This is given by:

```
$PROD:W s:SIGMA
O:PU Q:200
I:PX Q:80 P:1.25
I:PY Q:100
```

It should be clear that MPS/GE is going to do a lot of work for you, but you must give it the correct information to get a correct calibration.

The endogenous markup is set using two auxiliary variables and two constraint equations.

```
$CONSTRAINT:SHAREX

SHAREX =E= 80*PX*X / (80*PX*X + 100*PY*Y);

$CONSTRAINT:MARKUP

MARKUP =E= 1/(SIGMA - (SIGMA-1)*SHAREX);
```

The rest of the coding should be clear at this point.

\$TITLE Model M51-MPS.GMS: Closed 2x2 Economy, monopoly X producer

\* MPS/GE version

# \$ONTEXT

		Pro	Production Sectors			Consumers		
Markets		X	Y	W		CONS	ENTRE	
PX	!	100		-100				
PY			100	-100				
PU				200	Ì	-180	-20	
PW		-32	-60		İ	92		
PZ		-48	-40		İ	88		
MK	ĺ	-20			j		20	

## \$OFFTEXT

SCALAR SIGMA Elasticity of substitution,
INCOMEM Monopoly profit (in welfare units),
INCOMEC factor owners' income;

```
SIGMA = 9;
$ONTEXT
$MODEL:M51
$SECTORS:
                ! Activity level for sector X
        X
                ! Activity level for sector Y
        Y
        W
                ! Activity level for sector W (welfare)
$COMMODITIES:
                ! Price index for commodity X
        PX
                ! Price index for commodity Y
        PY
                ! Price index for factor L (net of tax)
```

! Price index for factor K

! Price index for welfare

PW

PZ

PU

### \$CONSUMERS:

CONS! Representative agent.

ENTRE ! Entreprenuer (monopolist)

### \$AUXILIARY:

SHAREX ! Value share of good X

MARKUP! X sector markup on marginal cost

\$PROD:X s:1

O:PX Q: 80 A:ENTRE N:MARKUP

I:PW Q: 32

I:PZ Q: 48

\$PROD:Y s:1

O:PY Q:100

I:PW Q:60

I:PZ Q:40

\$PROD:W s:SIGMA

O:PU 0:200

I:PX Q:80 P:1.25

I:PY Q:100

```
$DEMAND: CONS
       D:PU Q:180
       E:PW Q:92
       E:PZ Q:88
$DEMAND: ENTRE
  D:PU Q:20
$CONSTRAINT: SHAREX
        SHAREX =E= 80*PX*X / (80*PX*X + 100*PY*Y);
$CONSTRAINT: MARKUP
       MARKUP = E = 1/(SIGMA - (SIGMA-1)*SHAREX);
$OFFTEXT
$SYSINCLUDE mpsgeset M51
       Benchmark replication:
PX.L = 1.25;
SHAREX.L = 0.5;
MARKUP.L = 0.20;
```

```
PU.FX = 1;
$INCLUDE M51.GEN
SOLVE M51 USING MCP;
INCOMEM = W.L*(ENTRE.L/(ENTRE.L + CONS.L));
INCOMEC = W.L*(CONS.L/(ENTRE.L + CONS.L));
DISPLAY INCOMEM, INCOMEC;
        Evaluate the potential gains from first-best
*
        (marginalcost) pricing:
MARKUP.FX = 0;
$INCLUDE M51.GEN
SOLVE M51 USING MCP;
INCOMEM = W.L*(ENTRE.L/(ENTRE.L + CONS.L));
INCOMEC = W.L*(CONS.L/(ENTRE.L + CONS.L));
```

DISPLAY INCOMEM, INCOMEC;

# Model M52-MCP Monopoly and increasing returns in X, MCP version

One feature of model M51 is that it is not clear what is supporting monopoly power. The most plausible interpretation is that it is supported by some restrictive practice including a government licensing scheme. In the absence of restrictive practices, monopoly power is generally thought to be connected with increasing returns to scale in production. In model M52, we assume increasing returns to scale in X. More specifically, we assume the existence of fixed costs of production, and then constant marginal costs. This produces an average cost curve that is a rectangular hyperbole, with average cost approaching (asymptotic to) marginal cost for high levels of output.

While this is a convenient and tracktable technology, it does cause one difficulty, especially for MPS/GE. The code and indeed economic theory has difficulty with negative income for an agent in equilibrium. How is this to be squared with equilibrium and Walras' law? The simplest way around this is to just use one representative agent who both owns the factor endowment and receives markup revenues and pays fixed costs. Then back out monopoly profits once the model is solved. That is what I do here. The data matrix is a follows.

	Pro	Consumers			
Markets	X	N	Y	W	CONS
	•				·
				. – – – – – – –	
PX	100			-100	
PY	İ		100	-100	İ
PF	İ	20			-20
PU	ĺ			200	-200
PW	-32	-8	-60		100
PZ	-48	-12	-40		100
MK	-20				20

We are going to calibrate the model to zero profits initially. So markup revenues of 20 are entirely spent on fixed costs. Fixed costs are produced from unskilled and skilled labor. Since there is only one firm in this model, we do not specify a production or cost function for fixed cost, and it is treated as fixed and as a negative endowment for the representative consumer. The fixed costs appear in three equations of the model: demand for unskilled and skilled labor, and in the consumer's income constraint. Markup revenues are also added to consumers' income.

After solving the model, we recover monopoly profits. I specify this in welfare units, multiplying welfare (syntax W.L since we have already solved the model - recall that this is the way to report the value of a variable in an MCP or MPS/GE program). I multiply this by the ratio of the monopolist's profits, markup revenues minus fixed costs, to the total value of output. So the parameter INCOMEM gives the share of welfare that is monopoly profits. INCOMEC gives the income of consumers (factor owners).

MARKUP\*PX\*X\*80 - PZ\*12\*FCOST - PW\*8\*FCOST;

```
INCOMEM = W.L*((MARKUP.L*PX.L*X.L*80 - PW.L*8*FCOST -
PZ.L*12*FCOST)/(PX.L*X.L*80 + PY.L*Y.L*100));
```

```
INCOMEC = W.L - INCOMEM;
DISPLAY INCOMEM, INCOMEC;
```

The first counter-factual imposes marginal-cost pricing by setting the markup equal to zero (MARKUP.FX = 0). Welfare increases by 4% as in our previous example, but factor owners have a much bigger increase, assuming that they do not have to pay the deficit (an unrealistic assumption).

The second and third counterfactuals allow monopoly pricing and change the size of the economy. The correct syntax for re-freeing up the markup is to set upper and lower for the variable.

```
MARKUP.LO = -INF;
MARKUP.UP = INF;
```

Making the economy larger generates positive monopoly profits and making the economy smaller generates losses, even at the optimal markup. Thus we have a "natural monopoly" in this model. Note the influence of increasing returns in the experiment where we make the economy bigger. Real income and welfare increase more than in proportion to the size of the endowment. Welfare rises from 1.0 to 2.113; however, all of the increase goes to the monopolist and the welfare "per capita" of factor owners is

virtually unchanged at 1.998 after doubling the endowment.

\$TITLE Model M52-MCP.GMS: Closed Economy Monopoly with IRS \* uses MCP

# \$ONTEXT

Production Sectors								
Consumers Markets	X	N	Y	W	CONS			
PX	100			-100				
PY			100	-100				
PF		20			-20			
PU				200	-200			
PW	-32	-8	-60		100			
PZ	-48	-12	-40		100			
MK	-20				20			

### \$OFFTEXT

```
PARAMETERS
           Scale parameter for utility function,
А
           Elasticity of substitution,
SIGMA
          Monopoly profit (in welfare units),
INCOMEM
           Factor owners' income,
INCOMEC
          Endowment scale multiplier,
ENDOW
           Fixed costs scale multiplier,
FCOST
MODELSTAT;
SIGMA = 9;
ENDOW = 1;
FCOST = 1;
A = 0.5**(1/(1-SIGMA));
POSITIVE VARIABLES
X
Y
W
PX
```

```
PΥ
PU
PZ
PW
CONS
ENTRE
SHAREX
MARKUP;
EQUATIONS
DX
          Demand for X
DY
          Demand for Y
          Demand for W
DW
PRICEX
          MR = MC in X
          Zero profit condition for Y (PY = MC)
PRICEY
          Zero profit condition for W
PRICEW
          Supply-demand balance for skilled labor
SKLAB
          Supply-demand balance for unskilled labor
UNLAB
ICONS
          Consumer (factor owners') income
IENTRE
          Entrepreneur's profits
          Share of X in expenditure
SHX
          Markup equation;
MK
```

```
PRICEX.. (PW**0.40)*(PZ**0.60) = G = PX*(1 - MARKUP);
PRICEY.. (PW**0.60)*(PZ**0.40) = G = PY;
PRICEW.. A*((PX/1.25)**(1-SIGMA) +
           PY**(1-SIGMA))**(1/(1-SIGMA)) = G = PU;
           X*80 = E = A*(PX/1.25)**(-SIGMA)*
DX..
             ((PX/1.25)**(1-SIGMA) +
             PY**(1-SIGMA))**(SIGMA/(1-SIGMA))*W*200/1.25;
           Y*100 = E = A*PY**(-SIGMA)*
DY..
             ((PX/1.25)**(1-SIGMA) +
             PY**(1-SIGMA))**(SIGMA/(1-SIGMA))*W*200;
           W*200 = E = CONS/PU;
DW..
SKLAB..
           100 \times ENDOW = E =
              0.40*(PW**0.60)*(PZ**(0.40-1))*Y*100
            + 0.60*(PW**0.40)*(PZ**(0.60-1))*X*80 +12*FCOST;
```

```
UNLAB.. 100 \times ENDOW = E =
            0.60*(PW**(0.60-1))*(PZ**0.40)*Y*100
            + 0.40*(PW**(0.40-1))*(PZ**0.60)*X*80 + 8*FCOST;
ICONS.. CONS = E = PZ*100*ENDOW + PW*100*ENDOW +
            MARKUP*PX*X*80 - PZ*12*FCOST - PW*8*FCOST;
SHX.. SHAREX =E= 80*PX*X / (80*PX*X + 100*PY*Y);
MK..
          MARKUP = E = 1/(SIGMA - (SIGMA - 1)*SHAREX);
MODEL M52 /DX.PX, DY.PY, DW.PU, PRICEX.X, PRICEY.Y,
PRICEW.W, SKLAB.PZ, UNLAB.PW, ICONS.CONS,
SHX.SHAREX, MK.MARKUP/;
OPTION MCP=MILES;
OPTION LIMROW=0;
OPTION LIMCOL=0;
SOFFSYMLIST OFFSYMXREF OFFUELLIST OFFUELXREF
CONS.L = 200;
```

```
X.L = 1;
Y.L = 1;
W.L = 1;
PX.L = 1.25;
PY.L = 1;
PZ.L = 1;
PW.L = 1;
PU.L = 1;
CONS.L = 180;
ENTRE.L = 20;
SHAREX.L = 0.5;
MARKUP.L = 0.20;
PY.FX = 1;
*M52.ITERLIM = 0;
SOLVE M52 USING MCP;
MODELSTAT = M52.MODELSTAT - 1.;
INCOMEM = W.L*((MARKUP.L*PX.L*X.L*80 - PW.L*8*FCOST -
          PZ.L*12*FCOST) / (PX.L*X.L*80 + PY.L*Y.L*100));
```

```
INCOMEC = W.L - INCOMEM;
DISPLAY INCOMEM, INCOMEC;
* counterfactual: marginal-cost pricing
MARKUP.FX = 0;
SOLVE M52 USING MCP;
INCOMEM = W.L*((MARKUP.L*PX.L*X.L*80 - PW.L*8*FCOST -
          PZ.L*12*FCOST)/(PX.L*X.L*80 + PY.L*Y.L*100));
INCOMEC = W.L - INCOMEM;
DISPLAY INCOMEM, INCOMEC;
* counterfactual: double the size of the economy
MARKUP.LO = -INF;
```

```
MARKUP.UP = INF;
ENDOW = 2;
SOLVE M52 USING MCP;
INCOMEM = W.L*((MARKUP.L*PX.L*X.L*80 - PW.L*8*FCOST -
          PZ.L*12*FCOST) / (PX.L*X.L*80 + PY.L*Y.L*100));
INCOMEC = W.L - INCOMEM;
DISPLAY INCOMEM, INCOMEC;
* counterfactual: cut the size of the economy by 25%
ENDOW = 0.75;
SOLVE M52 USING MCP;
INCOMEM = W.L*((MARKUP.L*PX.L*X.L*80 - PW.L*8*FCOST -
          PZ.L*12*FCOST) / (PX.L*X.L*80 + PY.L*Y.L*100));
```

INCOMEC = W.L - INCOMEM;

DISPLAY INCOMEM, INCOMEC;

# Model M52-MPS Monopoly and increasing returns in X, MPSGE version

This model is quite similar to model M51, so there is not a lot to add concerning the MPS/GE version of model M52. The one thing worth noting is how the fixed costs are modeled here. These are specified as negative endowments in the consumer demand block. This is a good time to give a reminder that when a Q, P, or E field specifies an arithmetic operation the expression must be in parentheses. So an expression such as "Q: 100\*ENDOW" is not allowed.

```
$DEMAND: CONS
D: PU Q: 200
E: PW Q: (100*ENDOW)
E: PZ Q: (100*ENDOW)
E: PW Q: (-8*FCOST)
E: PZ Q: (-12*FCOST)
```

This formulation allows fixed costs and the overall endowment to be adjusted independently. The negative endowments are not a problem unless a consumer is left with negative income, a problem discussed earlier.

\$TITLE Model M52-MPS: Monopoly with IRTS - calibrated to zero profits

## \$ONTEXT

### Production Sectors

Consumers Markets	x	N	Y	W	CONS
PX	100			-100	
PY	İ		100	-100	
PF	İ	20			-20
PU	İ			200	-200
PW	-32	-8	-60		100
PZ	-48	-12	-40		100
MK	-20				20

\$OFFTEXT

# PARAMETERS Elasticity of substitution in demand, SIGMA FCOST Ratio of fixed costs to benchmark, ENDOW Level of factor endowment, INCOMEM Income of the monopolist, INCOMEC Inome of the factor owners; SIGMA = 9;FCOST = 1;ENDOW = 1;\$ONTEXT \$MODEL:M52 \$SECTORS: ! Activity level -- monopolist sector X Χ ! Activity level -- competitive sector Y ! Welfare index for the consumer W

### \$COMMODITIES:

PU! Welfare price index for the consumer

PX ! Price index for X (gross of markup)

PY ! Price index for Y (gross of markup)

PW ! Price index for labor

PZ ! Price index for capital

### \$CONSUMERS:

CONS ! All consumers

#### \$AUXILIARY:

SHAREX ! Value share of X in total consumption

MARKUP! Markup based on Marshallian demand

\$PROD:X s:1

O:PX Q: 80 A:CONS N:MARKUP

I:PW Q: 32

I:PZ Q: 48

\$PROD:Y s:1

O:PY Q:100

I:PW Q: 60

I:PZ Q: 40

\$PROD:W s:sigma

O:PU Q:200

I:PX Q: 80 P:1.25

I:PY Q:100

\$DEMAND: CONS

D:PU Q:200

E:PW Q: (100\*ENDOW)

E:PZ Q: (100\*ENDOW)

E:PW Q: (-8\*FCOST)

E:PZ Q: (-12\*FCOST)

\$CONSTRAINT: SHAREX

SHAREX\*(80\*PX\*X + 100\*PY\*Y) = G = 80\*PX\*X;

```
$CONSTRAINT: MARKUP
        MARKUP = E = 1/(SIGMA - (SIGMA-1)*SHAREX);
$OFFTEXT
$SYSINCLUDE mpsgeset M52
        Benchmark replication:
PX.L = 1.25;
SHAREX.L = 0.5;
MARKUP.L = 0.20;
$INCLUDE M52.GEN
SOLVE M52 USING MCP;
INCOMEM = W.L*((MARKUP.L*PX.L*X.L*80 - PW.L*8*FCOST -
          PZ.L*12*FCOST)/(PX.L*X.L*80 + PY.L*Y.L*100));
INCOMEC = W.L - INCOMEM;
DISPLAY INCOMEM, INCOMEC;
```

```
* counterfactual: marginal-cost pricing
MARKUP.FX = 0;
$INCLUDE M52.GEN
SOLVE M52 USING MCP;
INCOMEM = W.L*((MARKUP.L*PX.L*X.L*80 - PW.L*8*FCOST -
          PZ.L*12*FCOST)/(PX.L*X.L*80 + PY.L*Y.L*100));
INCOMEC = W.L - INCOMEM;
DISPLAY INCOMEM, INCOMEC;
* counterfactual: double the size of the economy
MARKUP.LO = -INF;
MARKUP.UP = INF;
ENDOW = 2;
```

```
$INCLUDE M52.GEN
SOLVE M52 USING MCP;
INCOMEM = W.L*((MARKUP.L*PX.L*X.L*80 - PW.L*8*FCOST -
          PZ.L*12*FCOST) / (PX.L*X.L*80 + PY.L*Y.L*100));
INCOMEC = W.L - INCOMEM;
DISPLAY INCOMEM, INCOMEC;
* counterfactual: cut the size of the economy by 25%
ENDOW = 0.75;
$INCLUDE M52.GEN
SOLVE M52 USING MCP;
```

DISPLAY INCOMEM, INCOMEC;

# Model M53-MCP Oligopoly, increasing returns, free entry in X, MCP version

Cases of pure monopoly or indeed cases with any fixed number of firms are likely rare in economics, and models allowing entry and exit as conditions change are clearly desirable. There is however a significant technical difficulty with allowing the number of firms to change but restricting the number to integer values. The equilibrium number of firms N is found when the N firms make non-negative profits but N+1 firms do make loses. The zero-profit condition cannot be specified as an equality when there are positive numbers of firms. Then the model become an integer complementarity problem and we have no good algorithms for solving this type of problem. The problem could be handled by brute force where there is only one type of firm by iterating over the (integer) number of firms, but even this breaks down when there are several possible firm types (e.g., firms from different countries, multinational versus domestic firms, etc.).

The classic and accepted way around this difficulty is to allow the number of firms to be a continuous variable. This is of course how monopolistic-competition models have always been formulated. While this trick is initially hard to swallow conceptually, it has the powerful analytical advantage of allowing us to specify the entry condition as an inequality, holding with equality if there are a positive number of firms active in equilibrium. This is the technique we adopt here. The data matrix for the problem is:

		Production	Sectors			Со	nsumers
Markets	X	N	Y	W		CONS	ENTRE
					·		
				100	 I		
PX	100			-100	ļ		
PY			100	-100			
PF		20					-20
PU				200		-200	
PW	-32	-8	-60			100	
PZ	-48	-12	-40			100	
MK	-20						20

Firm owners receive markup revenues and "demand" fixed costs. N will be an activity and in equilibrium the activity level will be interpreted as the number of firms producing X. The model is calibrated to zero profits initially, so the markup revenues (20) will cover fixed costs (20). Firm "owners" are represented by the single agent ENTRE.

Here I am going to introduce a somewhat different calibration scheme than that used above. Instead of assuming a given number of firms and using the data and markup formula to solve for the implied elasticity of substitution in demand, I am going to assume Cobb-Douglas preferences and solve for the implied number of firms in the

benchmark equilibrium. To do this, we need to derive the Cournot markup formula when there is more than one firm.

Revenue for a Cournot firm i is given by  $R_i = p(X_c)X_i$ . The Marshallian price elasticity of demand is denoted  $\eta$ .  $\eta$  is just -1 in our formulation with Cobb-Douglas demand in (6) since here  $\sigma = 1$ . Cournot conjectures imply that  $\partial X_c/\partial X_i = 1$ ; that is, a one-unit increase in own supply is a one-unit increase in market supply. Marginal revenue is then:

$$MR_{i} \equiv \frac{\partial R_{i}}{\partial X_{i}} = p + X_{i} \frac{\partial p}{\partial X_{c}} \frac{\partial X_{c}}{\partial X_{i}} = p + p \frac{X_{i}}{X_{c}} \left[ \frac{X_{c}}{p} \frac{\partial p}{\partial X_{c}} \right] \frac{\partial X_{c}}{\partial X_{i}} = p \left[ 1 - \frac{X_{i}}{X_{c}} \frac{1}{\eta} \right]$$

$$MR_i = p(1 - mk_i)$$
  $mk_i = \frac{s_i}{\eta}$   $\eta = -\left|\frac{p}{X_c}\frac{\partial X_c}{\partial p}\right|$   $\frac{X_i}{X_c} = s_i$ 

The combined assumptions of Cournot pricing and Cobb-Douglas preferences lead to a very simple expression for a firm's markup, which is given by the firm's market share.

$$mk_i = \frac{X_i}{X_c} \equiv s_i$$
 since  $\eta = 1$  (Cobb-Douglas)

In the data given above, the implied markup is 20%, thus there must be five firms active initially given Cobb-Douglas preferences. The model below is calibrated with N=5 initially. We now have two additional equations and unknowns relative to the previous model. There is the quantity (N) and unit price of fixed costs (PF) which are the complementary variables for the zero-profit equation for fixed costs (PRICEF) and the market clearing equation for fixed costs (DF) respectively. These are given by:

PRICEF...
$$(PW**0.40)*(PZ**0.60) = G= PF;$$
 (complementary to N)

DF... N\*4 = G= ENTRE/PF; (complementary to PF)

We have to remember that total fixed costs are 20 and therefore we must use N\*4 (= 20) for the demand for fixed costs for a consistent calibration at N = 5. N then appears in the markup equation as given by the algebra above and the markup revenue in turn provides the entrepreneur with income (ENTRE).

```
MK.. MARKUP*N =E= 1; (complementary to MARKUP)

IENTRE.. ENTRE =E= MARKUP*PX*X*80; (complementary to ENTRE)
```

Thus the entrepreneur is treated as a consumer who derives income from markup revenues with that income spent on a good called fixed costs. Total fixed costs are interpreted as the number of firms in equilibrium, which in turn feeds into the markup formula.

Alternatively, I could have chosen units to let N = 1, and then the correct quantity in the factor-demand equations and fixed-cost pricing equations would be N\*20 and the market equation would be MARKUP\*N\*5 =E=1. This latter case is convenient for the same reason that it is always convenient to have activity levels and prices calibrated to one initially: the new equilibrium levels in counter factual experiments are proportional changes from the benchmark. Many calibration errors are due to inconsistent use of units

and representative quantity and price units in different functions. So it is a useful exercise here to recalibrate the model to N = 1. N appears in four equations in the model and as just indicated, something must be changed in each of those four equations.

The counterfactual-experiment is to double the size of the economy. Note that this more than doubles welfare, which rises to 2.072. X production more than doubles but the number of firms much less than doubles, rising from 5.0 to 7.071. Increases in the number of firms leads each individual firm to perceive demand as more elastic and the equilibrium markup falls from 0.20 to 0.14. There is an increase in the output per firm, firms move down their average costs curves and become more technically efficient.

\$TITLE Model M53-MCP: Oligopoly with Free Entry, MCP \* version

## \$ONTEXT

### Production Sectors

Consumers					
Markets	X	N	Y	W	CONS ENTRE
·					·

PX	100			-100		
PY			100	-100	ĺ	
PF		20				-20
PU				200	-200	
PW	-32	-8	-60		100	
PZ	-48	-12	-40		100	
MK	-20					20

### \$OFFTEXT

#### PARAMETERS

SIGMA Elasticity of substitution,

```
Endowment scale multiplier,
ENDOW
MODELSTAT;
SIGMA = 1;
ENDOW = 1;
POSITIVE VARIABLES
X
Y
W
Ν
РX
PΥ
PU
PF
PZ
PW
CONS
ENTRE
MARKUP;
```

#### EQUATIONS

```
DX
         Demand for X
         Demand for Y
DY
DW Demand for W
DF Demand for fixed costs
PRICEX
         MR = MC in X
PRICEY
         Zero profit condition for Y (PY = MC)
PRICEW
         Zero profit condition for W
         Zero profit condition for fixed costs
PRICEF
SKTAB
         Supply-demand balance for skilled labor
         Supply-demand balance for unskilled labor
UNLAB
ICONS
         Consumer (factor owners') income
         Entrepreneur's profits
IENTRE
MK
         Markup equation;
PRICEX.. (PW**0.40)*(PZ**0.60) = G = PX*(1 - MARKUP);
PRICEY.. (PW**0.60)*(PZ**0.40) = G = PY;
PRICEW.. ((PX/1.25)**0.5)*(PY**0.5) = G = PU;
```

```
PRICEF.. (PW**0.40)*(PZ**0.60) = G = PF;
DX.. X*80 = E = 0.5*CONS/PX;
    Y*100 = E = 0.5*CONS/PY;
DY..
      W*200 = E = CONS/PU;
DW..
DF.. N*4 = G = ENTRE/PF;
SKLAB.. 100 \times ENDOW = E =
0.40*(PW**0.60)*(PZ**(0.40-1))*Y*100
            + 0.60*(PW**0.40)*(PZ**(0.60-1))*X*80
            + 0.60*(PW**0.40)*(PZ**(0.60-1))*N*4;
```

```
UNLAB.. 100 \times ENDOW = E =
            0.60*(PW**(0.60-1))*(PZ**0.40)*Y*100
            + 0.40*(PW**(0.40-1))*(PZ**0.60)*X*80
            + 0.40*(PW**(0.40-1))*(PZ**0.60)*N*4;
ICONS.. CONS =E = PZ*100*ENDOW + PW*100*ENDOW;
IENTRE.. ENTRE =E = MARKUP*PX*X*80;
MK.. MARKUP*N = E = 1;
MODEL M53 /DX.PX, DY.PY, DW.PU, DF.PF, PRICEX.X, PRICEY.Y,
PRICEW.W, PRICEF.N, SKLAB.PZ, UNLAB.PW, ICONS.CONS,
IENTRE.ENTRE, MK.MARKUP/;
OPTION MCP=MILES;
OPTION LIMROW=0;
OPTION LIMCOL=0;
SOFFSYMLIST OFFSYMXREF OFFUELLIST OFFUELXREF
CONS.L = 200;
```

```
X.L = 1;
Y.L = 1;
W.L = 1;
N.L = 4;
PX.L = 1.25;
PY.L = 1;
PZ.L = 1;
PW.L = 1;
PU.L = 1;
PF.L = 1;
CONS.L = 180;
ENTRE.L = 20;
MARKUP.L = 0.20;
PY.FX = 1;
*M53.ITERLIM = 0;
SOLVE M53 USING MCP;
MODELSTAT = M53.MODELSTAT - 1.;
```

\* counterfactual: double the size of the economy

ENDOW = 2;

SOLVE M53 USING MCP;

Model M53-MPC, Oligopoly, increasing returns, free entry in X, MPSGE version

Relative to M52, we now add one activity (N) and one commodity (PF) to the model. As in the MCP version just discussed, we need to be careful about calibration. If we choose N = 5 as the calibration, then the representative quantities for the \$PROD:N block are the total quantities from the data matrix divided by 5, and the markup equation just uses N.

```
$PROD:N s:1
O:PF Q: (20/5)
I:PW Q: (8/5)
I:PZ Q: (12/5)
$CONSTRAINT:MARKUP
MARKUP*N =E= 1;
```

If instead, we want an initial value of N = 1, then we would need

As I have emphasized several times, MPS/GE automatically generates the appropriate factor demand equations, and so there is no need to have to worry about consistency of units for these equations as we have to do in the MCP version.

\$TITLE Model M53-MPS: Oligopoly with free entry, MPS/GE
\* version

## \$ONTEXT

Production Sectors						Consumers		
Markets	X	N	Y	W		CONS	ENTRE	
РХ	100			-100				
PY			100	-100				
PF		20					-20	
PU				200		-200		
PW	-32	-8	-60			100		
PZ	-48	-12	-40			100		
MK	-20						20	

\$OFFTEXT

PARAMETERS

SIGMA

ENDOW;

```
SIGMA = 1;
ENDOW = 1;
SONTEXT
$MODEL:M53
$SECTORS:
 X ! Activity level - sector X output
   ! Activity level - competitive sector Y
 W ! Welfare index for the representative consumer
      ! Activity level - sector X fixed costs = no. of firms
$COMMODITIES:
         ! Price index for representative agent utility
  PU
          ! Price of good X (gross of markup)
  PX
          ! Price of good Y
  PY
          ! Unit price of inputs to fixed cost
  PF
         ! Price index for labor
  PW
  PZ
         ! Price index for capital
```

#### \$CONSUMERS:

CONS! Representative agent

ENTRE! Entrepreneur (converts markups to fixed cost)

### \$AUXILIARY:

MARKUP! Optimal markup based on Marshallian demand elas

\$PROD:X s:1

O:PX Q: 80 A:ENTRE N:MARKUP

I:PW Q: 32

I:PZ O: 48

\$PROD:Y s:1

O:PY 0:100

I:PW 0: 60

I:PZ Q: 40

\$PROD:N s:1

O:PF Q: (20/5)

I:PW Q: (8/5)

I:PZ Q: (12/5)

\$PROD:W s:1

O:PU Q:200

I:PX Q: 80 P:1.25

I:PY Q:100

\$DEMAND: CONS

D:PU Q:200

E:PW Q: (ENDOW\*100)

E:PZ Q: (ENDOW\*100)

\$DEMAND: ENTRE

D:PF 0: 20

\$CONSTRAINT: MARKUP

MARKUP\*N = E = 1;

\$OFFTEXT

\$SYSINCLUDE mpsgeset M53

\* Benchmark replication:

```
N.L = 5;
PX.L = 1.25;
MARKUP.L = 0.20;
$INCLUDE M53.GEN
SOLVE M53 USING MCP;
        Counterfactual double the size of economy.
*
ENDOW = 2;
$INCLUDE M53.GEN
SOLVE M53 USING MCP;
```

Model M54-MCP Two country version: Oligopoly, increasing returns and free entry in X, MCP version

Model M54 is a two-country version of M53. The initial data specifies that the countries are identical, and trade is costless. There is really not a lot new here, except that one has to be careful about the initial calibration of the markups. Markup revenues in the two markets are 20% of sales and once again we will assume Cobb-Douglas preferences. Thus there are once again 5 firms in total. This is turn implies that there are 2.5 firms in each country initially. I realize that this might be troublesome for some readers, and there are of course alternative calibrations such as the procedure discussed earlier where we choose the number of firms and calibrate the elasticity of substitution. Or in some cases research just use a "conjectural variation" parameter which is a multiplicative constant on the markup to reconcile the actual markup with the one given by the assumed number of firms. The modeler often has no alternative to this last procedure when trying to calibrate a model to real data.

A couple of features deserve discussion. First, this is a model in which I assume that markets are segmented, meaning that firms can price independently in the two markets (the two prices need not differ by exactly the transport/tariff cost). In general, firms will have different market shares and thus different markups in the two countries, and thus it is important to have different pricing conditions for the two supplies of a given firm.

Here I use a technique that is very useful in many models, which is to add additional activities. I first have one activity that converts factors into aggregate firm output. Then this output is the input into two separate activities, one which supplies output to the domestic market and one which supplies the foreign (export) market.

For country j, the PXDJ denotes the producer price, or cost of aggregate X output, XJ, so the value of this variable is the marginal cost of production. This aggregate supply is then divided by supply to j, XJJ, and exports to i, XJI (XJ = XJJ + XJI). Consumer price of XJ in country j is given by PXJ and that in country i by PXI. The relationship between marginal cost and consumer prices depends on the markups in the two markets and on transport costs on exported output. The relevant pricing equations for firms in j, with complementary variables in parenthesis are:

Given Cobb-Douglas preferences, a firm's markup is just its market share. I have coded the model such that the X variables just introduced are the total outputs of all firms

located in country j. Thus to get the output (sales) of an individual firm from j in a market, we have to divide these quantities by the number of j firms NJ. Thus, for example, the supply of a single j firm to market j is given by

```
(XJJ/NJ)/(XIJ + XJJ)) or XJJ/(NJ*(XIJ + XJJ));
```

The markup equations are thus given by:

```
MKII.. MARKII =E= XII/(NI*(XII + XJI));

MKIJ.. MARKIJ =E= XIJ/(NI*(XIJ + XJJ));

MKJJ.. MARKJJ =E= XJJ/(NJ*(XIJ + XJJ));

MKJI.. MARKJI =E= XJI/(NJ*(XII + XJI));
```

I coded the MPS/GE version of this model first, which took very little time. Then I tried to code this corresponding MCP. I made a few mistakes, things that MPS/GE does automatically, and it took me quite a while. So I pass on comments about my mistakes. First, trade costs require that more output must be shipped than is received. The relationships between firms' total outputs and supplies to the two markets are given by:

DXDI.. 
$$XII*40 + XIJ*40*(1+TC) = E = XI*80;$$

DXDJ.. 
$$XJJ*40 + XJI*40*(1+TC) = E = XJ*80;$$

Remember this when specifying factor demands. Second, remember the choice of units in correctly calibrating the functions. The markup equations given above are calibrated to NI = NJ = 2.5. Thus in the equations for demand for fixed costs and factor demands the 20 units shown in benchmark data must be written as N\*8 (=20). Unit are chosen such that variables such as YI, XII, and XIJ are all equal to one, and so quantity multiplies must be chosen that are consist with this choice of units. The correct market clearing equation for skilled labor in country i is:

```
SKLABI.. 100*ENDOWIS =E=
0.40*(WI**0.60)*(ZI**(0.40-1))*YI*100
+
0.60*(WI**0.40)*(ZI**(0.60-1))*(XII+XIJ*(1+TC))*40
+ 0.60*(WI**0.40)*(ZI**(0.60-1))*NI*8;
```

Study this equation and make sure you understand it: the first line is demand from Y, the second line is demand for factors used in marginal costs, and the third line is demand for factors in fixed costs.

Finally, one experiment is to put a subsidy on production in country i. The pricing

equation PRXDI is rather obvious. However, don't forget to charge the consumer for this subsidy. Country i's consumer's income is given by:

```
ICONSI.. CONSI =E= ZI*100*ENDOWIS + WI*100*ENDOWIL
- (WI**0.40)*(ZI**0.60)*SUBSIDY*XI*80;
```

This is of course another thing that MPS/GE does automatically for you. Note from the output that the subsidy (reminiscent of strategic trade policy) does not improve the welfare of country i, the potential benefits are dissipated by entry (Horstmann and Markusen, JIE 1986).

\$TITLE Model M54-MCP: Two-Country Oligopoly with free entry \* MCP version

## \$ONTEXT

	YI	YJ	XI	ХJ	NI	NJ	WI	ТМ	CONI	CONJ	EHTI	ENTJ
PYI	100						-100					
PYJ		100						-100				
PXI			100				-50	-50				
PXJ				100			-50	-50				
FCI					20						-20	
FCJ						20						-20
ZI	-40		-48		-12				100			
ZJ		-40		-48		-12				100		
WI	-60		-32		- 8				100			
WJ		-60		-32		- 8				100		
PUI							200		-200			
PUJ								200		-200		
MKI			-10	-10							10	10
MKJ			-10	-10							10	10

## \$OFFTEXT

#### **PARAMETERS**

```
ENDOWIL
 ENDOWIS
 ENDOWJL
 ENDOWJS
 TC
 SUBSIDY
MODELSTAT
 REALWI
 REALWJ
 REALZI
 REALZJ;
ENDOWIL = 1;
ENDOWIS = 1;
ENDOWJL = 1;
ENDOWJS = 1;
TC = 0;
SUBSIDY = 0;
```

#### POSITIVE VARIABLES

ΥI

ΥJ

WFI

WFJ

ΧI

XII

XIJ

ХJ

XJJ

XJI

NI

NJ

ΡY

PUI

PUJ

 $\mathtt{WI}$ 

 $\nabla J$ 

ZI

ZJ

PXI

PXJ

PXDI

**PXDJ** 

PFI

PFJ

CONSI

CONSJ

ENTI

ENTJ

MARKII

MARKIJ

MARKJI

MARKJJ;

#### **EQUATIONS**

DXDI X output in country i

DXI Demand for X in country i

DXDJ X output in country j

DXJ Demand for X in country j

DY Demand for Y

DWI Demand for welfare in country i

```
DMJ
             Demand for welfare in country j
             Demand for fixed costs in i
DFI
             Demand for fixed costs in j
DFJ
             Marginal cost of X in i
PRXDT
PRXII
             MR = MC \text{ for XII}
             MR = MC \text{ for XIJ}
PRXIJ
             Marginal cost of X in j
PRXDJ
             MR = MC \text{ for } Xjj
PRXJJ
             MR = MC \text{ for } Xji
PRXJI
PRYT
             Zero profits for YI
             Zero profits for YJ
PRYJ
             Zero profits for WFI
PRWT
             Zero profits for WFJ
PRWJ
             Zero profits for FI
PRFI
             Zero profits for FJ
PRFJ
             Market clearing for SI
SKLABI
             Market clearing for SJ
SKLABJ
             Market clearing for LI
UNLABI
             Market clearing for LJ
UNLABJ
ICONSI
             Consumer income in i
             Consumer income in j
ICONSJ
             Entreprenuer's income (markups) in i
IENTREI
```

```
IENTREJ Entrepreneur's income (markups) in j
           Markup ii
MKII
           Markup ij
MKIJ
           Markup jj
MKJJ
           Markup ji;
MKJI
PRXDI.. (WI**0.40)*(ZI**0.60)*(1-SUBSIDY) = G = PXDI;
PRXDJ... (WJ**0.40)*(ZJ**0.60) = G = PXDJ;
PRXII.. PXDI =G = PXI*(1 - MARKII);
PRXIJ.. PXDI*(1+TC) = G = PXJ*(1 - MARKIJ);
PRXJJ... PXDJ = G = PXJ*(1 - MARKJJ);
PRXJI.. PXDJ*(1+TC) = G = PXI*(1 - MARKJI);
PRYI.. (WI**0.60)*(ZI**0.40) = G = PY;
PRYJ.. (WJ**0.60)*(ZJ**0.40) = G = PY;
```

```
PRWI.. ((PXI/1.25)**0.5)*(PY**0.5) = G = PUI;
PRWJ... ((PXJ/1.25)**0.5)*(PY**0.5) =G= PUJ;
PRFI.. (WI**0.40)*(ZI**0.60) = G = PFI;
PRFJ... (WJ**0.40)*(ZJ**0.60) =G= PFJ;
DXDI.. XII*40 + XIJ*40*(1+TC) = E = XI*80;
DXDJ.. XJJ*40 + XJI*40*(1+TC) = E = XJ*80;
DXI.. (XII*40 + XJI*40) = E = 0.5*CONSI/PXI;
DXJ.. (XJJ*40 + XIJ*40) = E = 0.5*CONSJ/PXJ;
DY.. (YI + YJ)*100 = E = 0.5*(CONSI + CONSJ)/PY;
DWI.. WFI*200 =E= CONSI/PUI;
DWJ.. WFJ*200 = E = CONSJ/PUJ;
```

```
DFI.. NI*8 = G = ENTI/PFI;
```

DFJ.. 
$$NJ*8 = G = ENTJ/PFJ;$$

SKLABI.. 100\*ENDOWIS =E=

$$0.40*(WI**0.60)*(ZI**(0.40-1))*YI*100$$

+

SKLABJ..  $100 \times ENDOWJS = E =$ 

$$0.40*(WJ**0.60)*(ZJ**(0.40-1))*YJ*100$$

+

UNLABI.. 100\*ENDOWIL =E= 0.60\*(WI\*\*(0.60-1))\*(ZI\*\*0.40)\*YI\*100

```
0.40*(WI**(0.40-1))*(ZI**0.60)*(XII+XIJ*(1+TC))*40
          + 0.40*(WI**(0.40-1))*(ZI**0.60)*NI*8;
UNLABJ. 100*ENDOWJL = E =
0.60*(WJ**(0.60-1))*(ZJ**0.40)*YJ*100
0.40*(WJ**(0.40-1))*(ZJ**0.60)*(XJJ+XJI*(1+TC))*40
          + 0.40*(WJ**(0.40-1))*(ZJ**0.60)*NJ*8;
TCONST... CONST =E = ZT*100*ENDOWTS + WT*100*ENDOWTL
                   -(WI**0.40)*(ZI**0.60)*SUBSIDY*XI*80;
ICONSJ.. CONSJ = E = ZJ*100*ENDOWJS + WJ*100*ENDOWJL;
IENTREI.. ENTI =G= MARKII*PXI*XII*40 + MARKIJ*PXJ*XIJ*40;
IENTREJ.. ENTJ =G= MARKJJ*PXJ*XJJ*40 + MARKJI*PXI*XJI*40;
MKII.. MARKII =E = XII/(NI*(XII + XJI));
MKIJ.. MARKIJ =E=XIJ/(NI*(XIJ + XJJ));
```

```
MKJJ.. MARKJJ = E = XJJ/(NJ*(XIJ + XJJ));
MKJI.. MARKJI = E = XJI/(NJ*(XII + XJI));
MODEL M54 /DXDI.PXDI, DXDJ.PXDJ, DXI.PXI, DXJ.PXJ, DY.PY,
           DWI.PUI, DWJ.PUJ, DFI.PFI, DFJ.PFJ,
           PRXDI.XI, PRXII.XII, PRXIJ.XIJ,
           PRXDJ.XJ, PRXJJ.XJJ, PRXJI.XJI,
           PRYI.YI, PRYJ.YJ, PRWI.WFI, PRWJ.WFJ,
           PRFI.NI, PRFJ.NJ, SKLABI.ZI, SKLABJ.ZJ,
           UNLABI.WI, UNLABJ.WJ, ICONSI.CONSI, ICONSJ.CONSJ,
           IENTREI.ENTI, IENTREJ.ENTJ,
           MKII.MARKII, MKIJ.MARKIJ, MKJJ.MARKJJ,
MKJI.MARKJI/;
OPTION MCP=MILES;
OPTION LIMROW=0;
OPTION LIMCOL=0;
SOFFSYMLIST OFFSYMXREF OFFUELLIST OFFUELXREF
CONSI.L = 200;
CONSJ.L = 200;
```

```
ENTI.L = 20;
ENTJ.L = 20;
XI.L = 1;
XJ.L = 1;
XII.L = 1;
XIJ.L = 1;
XJJ.L = 1;
XJI.L = 1;
YI.L = 1;
YJ.L = 1;
WFI.L = 1;
WFJ.L = 1;
NI.L = 2.5;
NJ.L = 2.5;
PXDI.L = 1;
PXDJ.L = 1;
PXI.L = 1.25;
PXJ.L = 1.25;
PY.L = 1;
ZI.L = 1;
ZJ.L = 1;
WI.L = 1;
```

```
WJ.L = 1;
PUI.L = 1;
PUJ.L = 1;
PFI.L = 1;
PFJ.L = 1;
MARKII.L = 0.20;
MARKIJ.L = 0.20;
MARKJJ.L = 0.20;
MARKJI.L = 0.20;
PY.FX = 1;
*M54.ITERLIM = 0;
SOLVE M54 USING MCP;
MODELSTAT = M54.MODELSTAT - 1.;
* counterfactual: trade costs of 15%
TC = 0.15;
SOLVE M54 USING MCP;
```

```
* counterfactual: home production subsidy of 10%, trade
costs 0
TC = 0.;
SUBSIDY = .10;
SOLVE M54 USING MCP;
* counterfactual: country's identical except for size,
* positive trade costs (home market advantage)
SUBSIDY = 0;
TC = 0.15;
ENDOWIL = 1.5;
ENDOWJL = 0.5;
ENDOWIS = 1.5;
ENDOWJS = 0.5;
```

SOLVE M54 USING MCP;

```
REALWI = WI.L/PUI.L;
REALWJ = WJ.L/PUJ.L;
REALZI = ZI.L/PUI.L;
REALZJ = ZJ.L/PUJ.L;
DISPLAY REALWI, REALWJ, REALZI, REALZJ;
```

Model M54-MPS Two country version: Oligopoly, increasing returns and free entry in X, MCP version

As models get bigger, the advantages of MPS/GE become more and more obvious. It is not just the amount of coding saved by using MPS/GE, but the avoidance of errors due to the fact that MPS/GE automatically generates factor demands, income balance, and market clearing conditions. It took me about 40 minutes to code the MPS/GE version of this model, and about three or four hours to get the MCP version correct. Even then, I only knew that there were some errors in the MCP version because results from a couple of the counterfactuals did not match those from the MPS/GE version. I do think that it is very important for new modelers to code MCP versions of simple problems, so that the modeler understands exactly what MPS/GE is doing and exactly what square system is being generated in the background.

There is virtually nothing to add at this point. One thing that I might alert modelers to is possible divide by zero errors that I have had to deal with in my models of multinationals. When there are several types of firms, not all of which are active in equilibrium, there can be a divide-by-zero problem in the markup equations with NI zero if there are no type-NI firms in equilibrium. The simplest solution is just to set a lower bound on such a variable by using an expression like

```
NI.LO = 0.0001;
```

before the solve statement. This in not perfectly satisfactory since if that firm type is not active, some resources will still be consumed in fixed costs, but the approximation error is small. A more sophisticated solution to this difficulty is found in appendix 5 of my book on multinational firms ("Multinational Firms and the Theory of International Trade", MIT press, 2002).

A related problem is if there is an entrepreneur who is the sole demander of a particular (fixed cost) good and that firm type is not active, then there is no demand for that good. Then the price of that good is zero, which in turn causes the solution algorithm problems insofar as that prices may appear in the denominator of some expression in the solution routine. This problem only occurs in the initialization of the model and is easily solved. In the case of this model, this problem is easily solved by including the following statements before the solve statements to ensure that the initial values of incomes are not zero.

```
FCI.L = MAX(FCI.L, 0.0001);
FCJ.L = MAX(FCJ.L, 0.0001);
$INCLUDE M54.GEN
```

#### SOLVE M54 USING MCP;

Note again the difference between the suffixes ".L" which stands for "level", and sets the current value of a variable but allows the variable full freedom to change, and ".LO" which stands for "fix the lower bound of the variable at...".

\$TITLE: M54-MPS.GMS. Two country oligopoly model with free entry

\* uses MPS/GE

## \$ONTEXT

	YI	YJ	XI	ХJ	NI	NJ	WI	ŪИ	CONI	CONJ	EHTI	ENTJ
PYI	100						-100					
PYJ		100						-100				
PXI			100				-50	-50				
PXJ				100			-50	-50				
FCI					20						-20	
FCJ						20						-20
ZI	-40		-48		-12				100			
ZJ		-40		-48		-12				100		
WI	-60		-32		- 8				100			
WJ		-60		-32		-8				100		
PUI							200		-200			
PUJ								200		-200		
MKI			-10	-10							10	10
MKJ			-10	-10							10	10

#### \$OFFTEXT

# PARAMETERS ENDOWIL ENDOWJL ENDOWIS **ENDOWJS** REALWI REALWJ REALZI REALZJ TC SUBSIDY; TC = .0;SUBSIDY = 0;ENDOWIL = 1;ENDOWJL = 1;ENDOWIS = 1;ENDOWJS = 1;

# \$ONTEXT

\$MODEL:M54

# \$SECTORS:

YI YJ

WFI WFJ

ΧI

XII

XIJ

ХJ

XJI

XJJ

NI NJ

#### \$COMMODITIES:

PΥ

PUI PUJ

UW IW

ZI ZJ

PXI PXJ

PXDI

PXDJ

FCI FCJ

## \$CONSUMERS:

CONSI CONSJ

ENTI ENTJ

## \$AUXILIARY:

MARKII

MARKIJ

MARKJI

MARKJJ

```
$PROD:YI s:1.0
O:PY
        Q:100.0
I:WI
     Q:60.0
     Q:40.0
I:ZI
        s:1.0
$PROD:YJ
        Q:100.0
O:PY
I:WJ
     Q:60.0
I:ZJ
     Q:40.0
$PROD:XI
        s\!:\!1
O:PXDI
        Q:80.
      Q:32 A:CONSI S:SUBSIDY
I:WI
I:ZI
         Q:48 A:CONSI
                       S:SUBSIDY
$PROD:XII
O:PXI
      Q:40.
               A:ENTI
                          N:MARKII
       Q:40.
I:PXDI
$PROD:XIJ s:0.0
         Q:40.
O:PXJ
                 A:ENTI
                          N:MARKIJ
       Q: (40.*(1+TC))
I:PXDI
```

```
$PROD:XJ s:1
O:PXDJ Q:80.
I:WJ Q:32.
I:ZJ Q:48.
```

\$PROD:XJI

O:PXI Q:40. A:ENTJ N:MARKJI

I:PXDJ Q: (40.\*(1+TC))

\$PROD:XJJ s:0.0

O:PXJ Q:40. A:ENTJ N:MARKJJ

I:PXDJ Q:40.

\$PROD:NI s:1

O:FCI Q: (20/2.5)

I:WI Q: (8/2.5)

I:ZI Q:(12/2.5)

```
$PROD:NJ
       s:1
     Q:(20/2.5)
O:FCJ
I:WJ Q: (8/2.5)
    Q:(12/2.5)
I:ZJ
$PROD:WFI s:1.0
     Q:200.
O:PUI
I:PXI Q:80. P:1.25
I:PY
     Q:100.
$PROD:WFJ s:1.0
O:PUJ
      Q:200.
I:PXJ 0:80. P:1.25
     Q:100.
I:PY
$DEMAND: CONSI
D:PUI Q:200
E:WI
    Q: (100.*ENDOWIL)
```

Q: (100.\*ENDOWIS)

E:ZI

```
$DEMAND: CONSJ
D:PUJ Q:200
E:WJ Q:(100.*ENDOWJL)
E:ZJ Q:(100.*ENDOWJS)
$DEMAND: ENTI
D:FCI 0:20
$DEMAND: ENTJ
D:FCJ 0:20
$CONSTRAINT:MARKII
MARKII*NI*(XII + XJI) = G = XII;
$CONSTRAINT: MARKIJ
MARKIJ*NI*(XIJ + XJJ) = G = XIJ;
$CONSTRAINT:MARKJI
MARKJI*NJ*(XII + XJI) = G = XJI;
```

\$CONSTRAINT: MARKJJ MARKJJ\*NJ\*(XIJ + XJJ) = G = XJJ;

```
$OFFTEXT
$SYSINCLUDE MPSGESET M54
PXI.L = 1.25;
PXJ.L = 1.25;
MARKII.L = .2;
MARKJI.L = .2;
MARKIJ.L = .2;
MARKJJ.L = .2;
NI.L = 2.5;
NJ.L = 2.5;
PY.FX = 1.0;
NI.LO = 0.0001;
NJ.LO = 0.0001;
*OPTION SOLPRINT=OFF;
OPTION LIMROW=0;
```

```
OPTION LIMCOL=0;
$OFFSYMLIST OFFSYMXREF OFFUELLIST OFFUELXREF
$INCLUDE M54.GEN
SOLVE M54 USING MCP;
* counterfactual: trade costs of 15%
TC = 0.15;
$INCLUDE M54.GEN
SOLVE M54 USING MCP;
* counterfactual: home production subsidy of 10%, trade
costs 0
TC = 0.;
SUBSIDY = .10;
$INCLUDE M54.GEN
SOLVE M54 USING MCP;
```

```
* counterfactual: country's identical except for size,
* positive trade costs (home market advantage)
SUBSIDY = 0;
TC = 0.15;
ENDOWIL = 1.5;
ENDOWJL = 0.5;
ENDOWIS = 1.5;
ENDOWJS = 0.5;
$INCLUDE M54.GEN
SOLVE M54 USING MCP;
REALWI = WI.L/PUI.L;
REALWJ = WJ.L/PUJ.L;
REALZI = ZI.L/PUI.L;
REALZJ = ZJ.L/PUJ.L;
DISPLAY REALWI, REALWJ, REALZI, REALZJ;
```